

# Multilevel meta-analysis

Gabriela Ortiz

StataCorp LLC

October 3, 2023

Copyright ©2023 StataCorp LLC

# Overview

- Standard meta-analysis
- Multilevel meta-analysis
  - Fit a random-intercepts model
  - Fit a random-slope model
  - Assess heterogeneity at different levels
  - Perform sensitivity analysis

## Standard meta-analysis

# Meta-analysis

- This is a statistical technique for combining the results from several similar studies.
- The goal is to provide a single estimate of the effect of interest.
- If results vary widely across studies, the goal is then to understand the inconsistencies in the results.

# Student achievement data

- Consider a series of studies that examined whether students performed better under a modified school calendar, with frequent breaks, as opposed to the traditional schedule (Cooper et al. 2003).
- Each study was performed in a different school.
- The effect size is the standardized mean difference in performance.


# Modified school calendar data

```
. list study stdmdiff se if _n < 5 | _n > 52, sep(0) ab(20)
```

	study	stdmdiff	se
1.	1	-.18	.34351128
2.	2	-.22	.34351128
3.	3	.23	.37947332
4.	4	-.3	.37947332
53.	53	.12	.29495762
54.	54	.61	.28635642
55.	55	.04	.25884358
56.	56	-.05	.25884358

- Positive values indicate that students on the modified calendar performed better than students on the traditional calendar


# Visualizing the effect sizes



Heterogeneity:  $\tau^2 = 0.04$ ,  $I^2 = 58.09\%$ ,  $H^2 = 2.39$

Test of  $\theta_i = \theta_j$ :  $Q(9) = 21.67$ ,  $p = 0.01$

Test of  $\theta = 0$ :  $z = 1.31$ ,  $p = 0.19$



# Meta-analysis goals

- The department of education needs to decide whether they should implement the modified schedule
- Our goal is to report a single estimate of the standardized mean difference in performance
  - We'll assume that the effect sizes are independent across studies.
- If we observe substantial variation across the studies, we instead focus on trying to explain this variation
- Perhaps the design of the test, rigor of the academic curriculum, or some other study-level covariates can explain the discrepancies



# Random effects meta-analysis model

- K independent studies, each reports:
  - an estimate,  $\hat{\theta}_j$ , of the true (unknown) effect size  $\theta_j$
  - an estimate,  $\hat{\sigma}_j$ , of its standard error


$$\hat{\theta}_j = \theta + u_j + \epsilon_j$$

for  $j = 1, 2, \dots, K$ , where  $\epsilon_j \sim \mathcal{N}(0, \hat{\sigma}_j^2)$  and  $u_j \sim \mathcal{N}(0, \tau^2)$ .

The  $\epsilon_j$ s are the sampling errors and the  $u_j$ s are the random effects


- The estimate of the overall effect size is the mean of the distribution of effect sizes,  $\theta_{pop} = \mathbb{E}(\theta_j)$ .

# True difference in performance in the population




Suppose that the modified school calendar improves student performance by 20 points.

# Underlying values for each study



The true improvement for a study will vary from the population value due to teachers' experience, school resources, parental support, and other factors.

# Estimate of the difference for study 1



# Random effects

- The random effect,  $u_1$ , is the deviation of the study's true effect size from the overall mean,  $\theta$  (Borenstein et al., 2009)
- The sampling error,  $\epsilon_1$ , is the deviation of the observed effect size from the study's true effect size
- We're assuming the true effects are normally distributed, and we only have a sample of studies from the population of interest

# Standard meta analysis

```
. use schoolcal3, clear
(Effect of modified school calendar on student achievement)
```

```
. quietly: meta set stdmdiff se
```

```
. meta summarize, nostudies
```

```
Effect-size label: Effect size
```

```
Effect size: stdmdiff
```

```
Std. err.: se
```

```
Meta-analysis summary
```

```
Random-effects model
```

```
Method: REML
```

```
Number of studies =      56
```

```
Heterogeneity:
```

```
tau2 = 0.0884
```

```
I2 (%) = 94.70
```

```
H2 = 18.89
```

	Estimate	Std. err.	z	P> z	[95% conf. interval]	
theta	.1279321	.0438703	2.92	0.004	.0419479	.2139162

```
Test of homogeneity: Q = chi2(55) = 578.86
```

```
Prob > Q = 0.0000
```

## Multilevel meta-analysis


# Multilevel data

- Previously, we assumed that the effect sizes were independent across studies
- This assumption wouldn't be valid if the schools in our dataset were nested within districts, because the estimated effect size for schools within the same district would likely be correlated
- If our meta-analytic data have a multilevel (hierarchical) structure, we can perform multilevel meta-analysis to account for the correlation between effect sizes in the same group




# Standard meta-analysis as a two-level model

- Here we see the effect size reported by each study



# Three-level model

- Now suppose that multiple studies belong to the same district
- Schools belonging to the same district will be more similar in terms of demographics and socioeconomical factors, resulting in a correlation between results within a district



- Here we see how studies are grouped by district

# Modified school calendar data

```
. describe
```

Contains data from schoolcal3.dta

Observations: 56

Variables: 10

Effect of modified school calendar on student achievement

20 Sep 2023 12:02

(\_dta has notes)

Variable name	Storage type	Display format	Value label	Variable label
district	int	%12.0g		District ID
school	byte	%9.0g		School ID
study	byte	%12.0g		Study ID
stdmdiff	double	%10.0g		Standardized difference in means of achievement test scores
var	double	%10.0g		Within-study variance of stdmdiff
year	int	%12.0g		Year of the study
se	double	%10.0g		Within-study standard-error of stdmdiff
year_c	byte	%9.0g		Year of the study centered around 1990
mean_exp	float	%9.0g		Mean teacher experience
class_size	float	%9.0g		Mean class size

Sorted by: stdmdiff

# Modified school calendar data

```
. list district school study stdmdiff mean_exp in 1/11, sepby(district)
```

	district	school	study	stdmdiff	mean_exp
1.	11	1	1	-.18	6.394918
2.	11	2	2	-.22	1.820014
3.	11	3	3	.23	7.86858
4.	11	4	4	-.3	8.369441
5.	12	1	5	.13	10.48499
6.	12	2	6	-.26	10.73829
7.	12	3	7	.19	2.892403
8.	12	4	8	.32	6.689758
9.	18	1	9	.45	5.5483
10.	18	2	10	.38	13.40538
11.	18	3	11	.29	3.927117

# Multilevel meta-analysis model

By performing a multilevel meta-analysis, we can

- estimate the effect size more precisely by accounting for the dependence between observations within a group
- assess the heterogeneity between schools within a district and between districts
- estimate how each district varies from the overall mean
  - This will help us decide whether the modified calendar should be applied to some districts and not others


# Multilevel meta-analysis model

We'll fit a three-level random-intercepts model


$$\hat{\theta}_{jk} = \theta + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$

where  $u_j^{(3)} \sim \mathcal{N}(0, \tau_3^2)$ ,  $u_{jk}^{(2)} \sim \mathcal{N}(0, \tau_2^2)$ , and  $\epsilon_{jk} \sim \mathcal{N}(0, \hat{\sigma}_{jk}^2)$ . Note that  $j$  represents the third level (district),  $k$  represents the second level (school within district), and  $\epsilon_{jk}$  represents the sampling errors.

# True difference in performance in the population




## Schools within a district






# Schools and districts



# Estimate of the difference for study 1



# Syntax for multilevel random-intercepts meta-regression

- We can specify the effect-size standard errors

```
. meta multilevel depvar [ indepvars ] , relevels(levels)  
essevariable(varname)
```

- or the effect-size variances

```
. meta multilevel depvar [ indepvars ] , relevels(levels)  
esvarvariable(varname)
```

Option `relevels()` specifies the grouping structure, beginning with the highest level.

# Three-level meta-analysis

```
. meta multilevel stdmdiff, relevelevs(district school) essevariable(se) nolog
Multilevel REML meta-analysis
```

Number of obs = 56

Grouping information

Group variable	No. of groups	Observations per group		
		Minimum	Average	Maximum
district	11	3	5.1	11
school	56	1	1.0	1

Log restricted-likelihood = -7.9587239

Wald chi2(0) = .

Prob > chi2 = .

stdmdiff	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
_cons	.1847132	.0845559	2.18	0.029	.0189866	.3504397

Test of homogeneity: Q\_M = chi2(55) = 578.86

Prob > Q\_M = 0.0000

Random-effects parameters		Estimate
district: Identity		
	sd(_cons)	.2550724
school: Identity		
	sd(_cons)	.1809324

# Assess variability among effect sizes

```
. estat heterogeneity
```

Method: Cochran

Joint:

I2 (%) = 90.50

Method: Higgins–Thompson

district:

I2 (%) = 63.32

school:

I2 (%) = 31.86

Total:

I2 (%) = 95.19

# Higgins–Thompson heterogeneity statistics

- Higgins–Thompson  $I^2$  statistics for level 3 (district) and level 2 (school)

$$I_{District}^2 = \frac{\hat{\tau}_D^2}{\hat{\tau}_D^2 + \hat{\tau}_S^2 + s_{HT}^2}$$

$$I_{School}^2 = \frac{\hat{\tau}_S^2}{\hat{\tau}_D^2 + \hat{\tau}_S^2 + s_{HT}^2}$$

where  $\hat{\tau}_D^2$  and  $\hat{\tau}_S^2$  are the estimated variances for the district and school levels, and  $s_{HT}^2$  is the summary measure of the level 1 variances.

# Comparing the three and two level models

- Two-level model:

$$\hat{\theta}_j = \theta + u_j^{(2)} + \epsilon_j$$

- Three-level model:

$$\hat{\theta}_{jk} = \theta + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$

where  $u_j^{(3)} \sim \mathcal{N}(0, \tau_3^2)$  and  $u_{jk}^{(2)} \sim \mathcal{N}(0, \tau_2^2)$ .

- $j$  represents the third level (district)
- $k$  represents the second level (school within district)

# Comparing the three and two level models

- We want to test whether there is a nonnegligible amount of heterogeneity between the districts
- Essentially, we're testing whether  $\tau_3^2 = 0$



# Comparing the three and two level models

- First, we store our results from the previous model

```
. meta multilevel stdmdiff, ///  
  relevels(district school) essevariable(se)  
. estimates store full_model
```

- We now fit a two-level model with study as the second level

```
. meta multilevel stdmdiff, ///  
  relevels(study) essevariable(se)  
. estimates store standard_model
```

# Likelihood-ratio test

```
. lrtest full_model standard_model
```

Likelihood-ratio test

Assumption: standard\_model nested within full\_model

LR chi2(1) = 17.77

Prob > chi2 = 0.0000

Note: The reported degrees of freedom assumes the null hypothesis is not on the boundary of the parameter space. If this is not true, then the reported test is conservative.

Note: LR tests based on REML are valid only when the fixed-effects specification is identical for both models.

- We prefer the full model with district and school-level random effects

# Fit a two-level model

- We want to test whether there is a nonnegligible amount of heterogeneity between the schools within a district
- First, we store our results from the previous model

```
. meta multilevel stdmdiff, ///  
  relevels(district school) essevariable(se)  
  estimates store full_model
```
- We now fit a two-level model with district as the second level

```
. meta multilevel stdmdiff, ///  
  relevels(district) essevariable(se)  
  estimates store district_re
```

# Likelihood-ratio test

```
. lrtest full_model district_re
```

Likelihood-ratio test

Assumption: district\_re nested within full\_model

LR chi2(1) = 48.52

Prob > chi2 = 0.0000

Note: The reported degrees of freedom assumes the null hypothesis is not on the boundary of the parameter space. If this is not true, then the reported test is conservative.

Note: LR tests based on REML are valid only when the fixed-effects specification is identical for both models.

# Estimation method

- We can fit our models using maximum likelihood (ML) or restricted maximum likelihood (REML).
- Restricted maximum likelihood is also known as residual maximum likelihood, and it is the default method because it produces unbiased estimates.
- Note that for REML the likelihood does not depend on the fixed-effects component of the model.

# Information criteria

```
. estimates stats full_model district_re standard_model, n(55)
```

Akaike's information criterion and Bayesian information criterion

Model	N	ll(null)	ll(model)	df	AIC	BIC
full_model	55	.	-7.958724	3	21.91745	27.93945
district_re	55	.	-32.21648	2	68.43295	72.44762
standard_m~l	55	.	-16.8455	2	37.691	41.70566

# Preparing table of estimation results

```
. etable, estimates(full_model district_re standard_model)
```

	stdmdiff	stdmdiff	stdmdiff
Intercept	0.185 (0.085)	0.196 (0.090)	0.128 (0.044)
sd(_cons)	0.255 (0.070)	0.288 (0.068)	
sd(_cons)	0.181 (0.031)		
(output omitted)			
sd(_cons)			0.297 (0.034)
Number of observations	56	56	56

# Customizing and exporting the table

```
. etable, estimates(full_model district_re standard_model) ///
> equations(stdmdiff district school) showeq ///
> mstat(N) mstat(aic) mstat(bic) column(index) ///
> export(mytable.docx)
```

	1	2	3
Standardized difference in means of achievement test scores			
Intercept	0.185 (0.085)	0.196 (0.090)	0.128 (0.044)
District ID			
sd(_cons)	0.255 (0.070)	0.288 (0.068)	
School ID			
sd(_cons)	0.181 (0.031)		
Number of observations	56	56	56
AIC	21.92	68.43	37.69
BIC	27.99	72.48	41.74

(collection ETable exported to file mytable.docx)



# Sensitivity analysis

- Suppose we're interested in exploring how different magnitudes of the school-level variation impact our estimates of the standardized mean difference and the district-level variation.
- To answer this question, we'll refit our model, each time setting the random-effects standard deviations for the school level to a different value.

# Random-intercepts standard deviations

```
. meta multilevel stdmdiff, ///  
  releveles(district school, sd(. 0.01)) esse(se)  
. estimates store fixsd1  
  
. meta multilevel stdmdiff, ///  
  releveles(district school, sd(. 0.18)) esse(se)  
. estimates store fixsd2  
  
. meta multilevel stdmdiff, ///  
  releveles(district school, sd(. 0.60)) esse(se)  
. estimates store fixsd3
```

# Comparing effect sizes

```
. estimates table _all, stats(sd2) keep(stdmdiff:_cons) b(%8.3f) se(%8.3f)
```

Variable	fixsd1	fixsd2	fixsd3
_cons	0.196	0.185	0.123
	0.090	0.085	0.083
sd2	0.010	0.180	0.600

Legend: b/se

# Accessing stored random-effects standard deviations

```
. meta multilevel stdmdiff, relevels(district school, sd(. 0.01)) ///
> essevariable(se) nogroup nolog noheader nofetable
```

Random-effects parameters	Estimate
district: Identity sd(_cons)	.2876359
school: Custom sd(_cons)	.01*

(\*) fixed during estimation

```
. matrix list e(b)
```

```
e(b)[1,3]
```

```
      stdmdiff:   lns1_1_1:   lnsig_e:
             _cons      _cons      _cons
y1   .19597049  -1.2460597         0
```

```
. display exp(-1.2460597)
.28763594
```

# Comparing random-effects standard deviations for districts

```
. estimates table _all, stats(sd2) keep(lns1_1_1:_cons) b(%8.3f) eform
```

Variable	fixsd1	fixsd2	fixsd3
_cons	0.288	0.255	0.000
sd2	0.010	0.180	0.600


# Predictions of random effects

```
. qui: meta multilevel stdmdiff, relevels(district school) esse(se)  
. predict double u3 u2, reffects reses(se_u3 se_u2, diagnostic)  
. by district, sort: generate tolist = (_n==1)  
. list district u3 se_u3 if tolist
```

	district	u3	se_u3
1.	11	-.18998595	.07071818
5.	12	-.08467077	.13168501
9.	18	.1407273	.11790486
12.	27	.24064814	.13641505
16.	56	-.1072942	.13633364
20.	58	-.23650899	.15003184
31.	71	.5342778	.12606073
34.	86	-.2004695	.1499012
42.	91	.05711692	.14284823
48.	108	-.14168396	.13094894
53.	644	-.01215679	.10054689

# Normal quantile plot

```
. generate double uстан3 = u3/se_u3
. qnorm uстан3 if tolist, mlabel(district)
```



# Three-level meta-analysis and meta-regression

- Multilevel meta-analysis; random intercepts for district (3) and school (2):

$$\hat{\theta}_{jk} = \theta + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$



# Three-level meta-analysis and meta-regression

- Multilevel meta-analysis; random intercepts for district (3) and school (2):

$$\hat{\theta}_{jk} = \theta + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$

- Multilevel meta-regression; random intercepts and a moderator

$$\hat{\theta}_{jk} = \beta_0 + \beta_1 \text{exper}_{jk} + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$

- One general effect of `exper` for all studies

# Three-level meta-analysis and meta-regression

- Multilevel meta-analysis; random intercepts for district (3) and school (2):

$$\hat{\theta}_{jk} = \theta + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$

- Multilevel meta-regression; random intercepts and a moderator


$$\hat{\theta}_{jk} = \beta_0 + \beta_1 \text{exper}_{jk} + u_j^{(3)} + u_{jk}^{(2)} + \epsilon_{jk}$$

- One effect of `exper` for all studies
- Multilevel meta-regression with random slopes


$$\hat{\theta}_{jk} = \beta_0 + \beta_1 \text{exper}_{jk} + u_{0j}^{(3)} + u_{1j}^{(3)} \text{exper}_{jk} + u_{jk}^{(2)} + \epsilon_{jk}$$

- One effect of `exper` specific to each district


# Multilevel meta-analysis




# Multilevel meta-regression



# Multilevel meta-regression



# Multilevel meta-regression with a random slope



# Models with random slopes

- To fit models with random slopes we need `meta meregress`

```
. meta meregress stdmdiff x1 || district: x1 || school:, esse(se)
```

- The random slope for `x1` allows the effect of `x1` to vary across districts

- Just like with `meta multilevel`, we can specify the effect-size standard errors

```
. meta meregress depvar fe_equation || re_equation  
[ || re_equation ] , essevariable(varname)
```

- or the effect-size variances

```
. meta meregress depvar fe_equation || re_equation  
[ || re_equation ] , esvarvariable(varname)
```

# Three-level meta-regression with random slopes

```
. meta meregress stdmdiff mean_exp ///
> || district: mean_exp ///
> || school:, essevariable(se) nolog nogroup
```

Multilevel REML meta-regression

Log restricted-likelihood = -3.3635425

Number of obs = 56  
Wald chi2(1) = 8.37  
Prob > chi2 = 0.0038

stdmdiff	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
mean_exp	-.0262054	.009058	-2.89	0.004	-.0439587	-.0084521
_cons	.3580009	.0981127	3.65	0.000	.1657036	.5502983

Test of homogeneity: Q\_M = chi2(54) = 558.47

Prob > Q\_M = 0.0000

Random-effects parameters	Estimate
district: Independent	
sd(mean_exp)	.0156308
sd(_cons)	.2605429
school: Identity	
sd(_cons)	.146955




# Display variance components

```
. estat sd, variance
```

Random-effects parameters	Estimate
district: Independent	
var(mean_exp)	.0002443
var(_cons)	.0678826
school: Identity	
var(_cons)	.0215958

# Checking model fit

- . predict double fit, fitted
- . twoway (scatter fit stdmdiff) (function y = x, range(stdmdiff))



# Random-effects covariance structures

$$\text{unstructured } \Sigma = \begin{bmatrix} \sigma_{11} & & \\ \sigma_{21} & \sigma_{22} & \\ \sigma_{31} & \sigma_{32} & \sigma_{33} \end{bmatrix}$$

$$\text{independent } \Sigma = \begin{bmatrix} \sigma_{11} & & \\ 0 & \sigma_{22} & \\ 0 & 0 & \sigma_{33} \end{bmatrix}$$

$$\text{exchangeable } \Sigma = \begin{bmatrix} \sigma_{11} & & \\ \sigma_{21} & \sigma_{11} & \\ \sigma_{21} & \sigma_{21} & \sigma_{11} \end{bmatrix}$$

$$\text{identity } \Sigma = \begin{bmatrix} \sigma_{11} & & \\ 0 & \sigma_{11} & \\ 0 & 0 & \sigma_{11} \end{bmatrix}$$

# Specifying the random-effects covariance structure

```
. meta meregress stdmdiff mean_exp ///
> || district: mean_exp, covariance(exchangeable) ///
> || school:, essevariable(se) nolog nogroup
```

Multilevel REML meta-regression

Number of obs = 56  
Wald chi2(1) = 2.62  
Prob > chi2 = 0.1054

Log restricted-likelihood = -10.569892

stdmdiff	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
mean_exp	-.0203673	.0125786	-1.62	0.105	-.0450208	.0042862
_cons	.3051209	.0695575	4.39	0.000	.1687908	.4414511

Test of homogeneity: Q\_M = chi2(54) = 558.47

Prob > Q\_M = 0.0000

Random-effects parameters	Estimate
district: Exchangeable	
sd(mean_exp _cons)	.0263766
corr(mean_exp, _cons)	.9999999
school: Identity	
sd(_cons)	.2195733

# Exchangeable covariance structure

```
. meta meregress stdmdiff mean_exp class_size ///
> || district: mean_exp class_size, covariance(exchangeable) ///
> || school:, essevariable(se) var matlog nolog nogroup noheader
```

stdmdiff	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
mean_exp	-.0132714	.0042179	-3.15	0.002	-.0215384	-.0050044
class_size	.0251681	.0029628	8.49	0.000	.0193611	.0309751
_cons	-.3344917	.0521791	-6.41	0.000	-.4367608	-.2322226

Test of homogeneity:  $Q_M = \chi^2(53) = 126.71$

Prob >  $Q_M = 0.0000$

Random-effects parameters	Estimate
district: Exchangeable	
var(mean_exp.._cons)(1)	.0000461
cov(mean_exp.._cons)(1)	-.000023
school: Identity	
var(_cons)	8.15e-09

(1) mean\_exp class\_size \_cons

# Random-effects covariance matrix

```
. estat recovariance
```

Random-effects covariance matrix for level district

	mean_exp	class_s~e	_cons
mean_exp	.0000461		
class_size	-.000023	.0000461	
_cons	-.000023	-.000023	.0000461

Random-effects covariance matrix for level school

	_cons
_cons	8.15e-09

## Summary

# Summary

- Today, we learned how to do the following in Stata:
  - Perform meta-regression with effect sizes that have hierarchical structures.
  - Assess heterogeneity at different levels of the hierarchy.
  - Create and export a table of estimation results.



# Perform meta-analysis using Stata's graphical interface

meta - Meta-Analysis Control Panel

Display meta settings    Modify meta settings

Setup

Summary

Forest plot

Heterogeneity

Regression

Publication bias

Multivariate

Multilevel

Note: Multivariate and multilevel meta-analyses do not require any setup. Proceed to the respective pane: Multivariate or Multilevel.

Declare meta-analysis data

☒ Compute and declare effect sizes for two-group comparison of continuous outcomes

☐ Compute and declare effect sizes for two-group comparison of binary outcomes

☐ Compute and declare effect sizes for estimating a single proportion (prevalence)

☐ Declare generic, precomputed effect sizes (in the metric closest to normality)

Main    If/in    Model    Options

Specify group 1 (treatment) variables

Sample size:  Mean:  Standard deviation:

Specify group 2 (control) variables

Sample size:  Mean:  Standard deviation:

Specify effect size

Effect size:

☐ Use exact computation for the bias-correction factor

☐ Use Hedges and Olkin standard error for effect size

Submit

No. of studies: <none>    Model: <none>    Effect size: <none>

CI level: <none>    Method: <none>    Std. err.: <none>

Close

# Resources

- Overview of **meta-analysis features** in Stata
- Video tutorial on **performing meta-analysis in Stata**
- **Stata Meta-Analysis Reference Manual**

# References

- Berkey, C. S., D. C. Hoaglin, F. Mosteller, and G. A. Colditz. 1995. A random effects regression model for meta analysis. *Statistics in Medicine* 14: 395–411.
- Borenstein, M., L. V. Hedges, J. P. T. Higgins, and H. R. Rothstein. 2009. *Introduction to Meta-Analysis*. Chichester, UK: Wiley.
- Colditz, G. A., T. F. Brewer, C. S. Berkey, M. E. Wilson, E. Burdick, H. V. Fineberg, and F. Mosteller. 1994. Efficacy of BCG vaccine in the prevention of tuberculosis: Meta analysis of the published literature. *Journal of the American Medical Association* 271: 698–702.
- Glass, G. V. 1976. Primary, secondary, and meta-analysis of research. *Educational Researcher* 5: 3–8.
- Moore, R. A., M. R. Tramèr, D. Carroll, P. J. Wiffen, and H. J. McQuay. 1998. Quantitative systematic review of topically applied non-steroidal anti-inflammatory drugs. *British Medical Journal* 316: 333–338.
- Rabe-Hesketh, S., and A. Skrondal. 2022. *Multilevel and longitudinal modeling using Stata*. 4th ed. College Station, TX: Stata Press.